# $Id$
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=libsamplerate
pkgver=0.1.9
pkgrel=1
pkgdesc="Secret Rabbit Code - aka Sample Rate Converter for audio"
arch=('i686' 'x86_64')
url="http://www.mega-nerd.com/SRC/index.html"
license=('BSD')
depends=('libsndfile')
source=(http://www.mega-nerd.com/SRC/libsamplerate-${pkgver}.tar.gz)
sha1sums=('ed60f957a4ff87aa15cbb1f3dbd886fa7e5e9566')
validpgpkeys=('73571E85C19F4281D8C97AA86CA41A7743B8D6C8')

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr
  make
}

check() {
  cd ${pkgname}-${pkgver}
  make check
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install

  install -Dm644 COPYING "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
}
